Points anguleux, de rebroussement et de tangence verticale

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue sous la version Maple 6. Il s'agit, dans ce document, d'étudier certains cas de non dérivabilité en $x = x_0$ d'une fonction continue en $x = x_0$ et leur interprétation graphique. Ce document aborde la distinction à faire entre un point anguleux et un point de rebroussement. De plus, ce document apporte au lecteur une maîtrise accrue du logiciel Maple.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
> with(plots,display,setoptions):
  Fond:=ColorTools:-Color([1,0.99215686,0.96862745]);
  setoptions(axesfont=[times,roman,8],size=[300,300]):
```

$$Fond := \langle \text{RGB} : 1\ 0.992\ 0.969 \rangle \tag{1.1}$$

Réglage de l'affichage pour les variables conditionnées

```
> with(Typesetting):
  interface(typesetting=standard); #niveau de composition Maple
  Standard
  interface(showassumed=2);        # Variables avec suppositiosnOhrase
```

$$extended, [extended]$$

$$2 \tag{1.2}$$

Point anguleux

Soit la fonction f définie par $f(x) = |1 - 2x| - x$. Le domaine de la fonction f est $\mathbb{R}$. Traçons son graphique pour $x \in [-1, 2]$.

```
> f:=x->abs(1-2*x)-x;
  plot([x,f(x),x=-1..2],view=[-1..2,-1..3],size=[300,300]);
```

$$f := x \rightarrow |1 - 2x| - x$$



Calculons $f\,'\left(\frac{1}{2}\right)$.

```
> D(f)(1/2);
Error, (in simpl/abs) abs is not differentiable at 0
```

L'afficheur vous informe que la fonction f n'est pas dérivable (différentiable) en $x = \frac{1}{2}$. Vous allez en découvrir la raison.

Calculons donc les dérivées à gauche $f'_{-}\left(\frac{1}{2}\right)$ et à droite $f'_{-}\left(\frac{1}{2}\right)$ de la fonction f. Pour cela, saisissons les points $P\left(\frac{1}{2}, f\left(\frac{1}{2}\right)\right)$ et $Q\left(\frac{1}{2}+h, f\left(\frac{1}{2}+h\right)\right)$ afin de pouvoir appliquer les définitions de la dérivée à droite et de la dérivée à gauche.

```
> P:=[1/2,f(1/2)];
  Q:=[P[1]+h,f(P[1]+h)];
  TVM:=(P,Q)->(Q[2]-P[2])/(Q[1]-P[1]);
  `f '`[`-`](1/2)=Limit(TVM(P,Q),h=0,left);
```

$$P := \left[\frac{1}{2}, -\frac{1}{2}\right]$$

$$Q := \left[\frac{1}{2}+h, 2\,|h| - \frac{1}{2} - h\right]$$

$$TVM := (P, Q) \rightarrow \frac{Q_2 - P_2}{Q_1 - P_1}$$

$$f'_{-}\left(\frac{1}{2}\right) = \lim_{h \rightarrow 0^-} \frac{2\,|h| - h}{h} \qquad \textbf{(2.1)}$$

Reste à évaluer cette limite

```
> value((2.1));
```

$$f'_{-}\left(\frac{1}{2}\right) = -3 \qquad \textbf{(2.2)}$$

Alors, la dérivée à gauche $f'_{-}\left(\frac{1}{2}\right)$ existe et elle vaut -3. Calculons maintenant la dérivée à droite en $x = \frac{1}{2}$.

```
> `f '`[`-`](1/2)=Limit(TVM(P,Q),h=0,right);
```

```
value(%);
```

$$f'_{`-`}\left(\frac{1}{2}\right) = \lim_{h\to 0+} \frac{2|h| - h}{h}$$

$$f'_{`-`}\left(\frac{1}{2}\right) = 1 \qquad \textbf{(2.3)}$$

La dérivée à droite $f'_{+}\left(\frac{1}{2}\right)$ existe elle aussi et elle vaut 1.

Mais puisque $f'_{-}\left(\frac{1}{2}\right) = -3 \neq 1 = f'_{+}\left(\frac{1}{2}\right)$, voilà la raison pour laquelle la dérivée $f\,'\left(\frac{1}{2}\right)$ n'existe pas.

Les développements précédents est l'occasion de se rappeler le TVI. Nous aurions pu obtenir le même résultat en évaluant les limites directionnelles gauche et droite en $x=\frac{1}{2}$ de la fonction dérivée f ' (Ces deux calculs de limites ne sont possibles seulement si la fonction f est dérivable dans n'importe quel voisinage centré en $x=\frac{1}{2}$.

```
> `f '`[`-`](1/2)=limit(D(f)(x),x=1/2,left);
  `f '`[`+`](1/2)=limit(D(f)(x),x=1/2,right);
```

$$f'_{`-`}\left(\frac{1}{2}\right) = -3$$

$$f'_{`+`}\left(\frac{1}{2}\right) = 1 \qquad \textbf{(2.4)}$$

On dit, du point $P\left(\frac{1}{2}, f\left(\frac{1}{2}\right)\right)$ qu'il est un point anguleux.

DÉFINITION

Un point $P(a, f(a))$, sur le graphique d'une fonction f, est un point anguleux si f est continue en $x = a$ et si l'une des deux conditions suivantes est satisfaite:

- les deux dérivées directionnelles existent en $x = a$ mais sont différentes.
- une seule des deux dérivées directionnelles existe et l'autre est telle que $|f'(x)| \to \infty$ lorsque $x \to a^{`+`}$ ou $x \to a^{`-`}$.

D'un point de vue graphique, en un point anguleux, il n'est alors pas possible de parler de la pente de « l'unique droite tangente » en ce point.

REMARQUE

L'exemple précédent nous a montré qu'il ne suffit pas qu'une fonction f soit continue en $x = a$ pour qu'elle puisse être dérivable en $x = a$. Autrement dit, la continuité est une condition nécessaire à la dérivabilité mais non suffisante.

En résumé, une fonction f continue en $x = a$ n'est pas nécessairement dérivable en $x = a$.

Nous savons maintenant que la fonction f n'est pas dérivable en $x=\frac{1}{2}$. Mais comment l'évaluateur dérivera la fonction f pour x quelconque ? Pour le découvrir, calculons avec l'opérateur de dérivation D, la dérivée de la fonction f.

```
> Diff(f(x),x)=D(f)(x);
```

$$\frac{d}{dx}\left(\left|-1+2x\right|-x\right)=2\,\mathrm{abs}(1,\,-1+2x)-1 \qquad \mathbf{(2.5)}$$

Le résultat affiché pour la dérivée de la fonction f est formulé avec une valeur absolue. C'est de cette manière que l'afficheur signifie que la dérivée de $\left|-1+2x\right|-x$ est *conditionnelle au signe algébrique* de la simplification de la dérivée $\mathrm{abs}(1,\,-1+2x)$. C'est clair pour Maple. Mais, pour nous, cette formulation algébrique ne l'est pas du tout.

Puisque la valeur de $1-2x$ change de signe autour de $x=\frac{1}{2}$, calculons donc la dérivée de la fonction f pour $x<\frac{1}{2}$ d'une part et, d'autre part, pour $x>\frac{1}{2}$.

La macro-commande `assume` permet de conditionner une variable qui va nous permettre par la suite de faire des simplifications conditionnelles.

```
> assume(x<1/2);
  Diff(f(x),x)=D(f)(x);
```

$$\frac{d}{dx}(1-3x)=-3$$

With assumptions on x **(2.6)**

```
> assume(x>1/2);
  Diff(f(x),x)=D(f)(x);
```

$$\frac{d}{dx}(-1+x)=1$$

With assumptions on x **(2.7)**

La macro-commande `assume` a imposé à l'évaluateur d'opérer la simplification du calcul de la dérivée sous certaines conditions, la dernière étant de considérer $x>\frac{1}{2}$. En effet, l'afficheur, en donnant la réponse du calcul de chaque dérivée, a affiché la phrase: *With assumptions on x*. C'est une façon de rappeler à l'usager que le calcul a été simplifié conditionnellement. Au lieu de cette phrase, l'autre façon pour l'afficheur de nous le rappeler aurait été de faire suivre, dans la réponse, la variable x par le caractère tilde ~. Ici, c'est le choix *Phrase* qui a été sélectionné dans le sous-menu *Variables avec suppositions* du menu de l'onglet *Affichage* du sous-menu *Options*... du menu *Outils* Il faut que le *Niveau de composition* soit *Standard Maple*.

Outils → *Options*... → *Affichage* → *Variables avec suppositions* et
Outils → *Options*... → *Affichage* → *Niveau de composition (Maple Stamdard)*

Maple étant un programme interactif, il faut donc, avant de poursuivre, rendre à nouveau la variable x libre soit, « inconditionnellement mathématique ».

```
> x:='x';
```

$$x:=x \qquad \mathbf{(2.8)}$$

Obtenons, d'une autre manière, la formulation algébrique de la dérivée $f'(x)$. Les deux dérivations précédentes nous a montré que la fonction f à dériver, définie avec une valeur absolue, est en fait une fonction définie par morceaux. Convertissez donc la règle f(x) en une règle définie par morceaux à l'aide de la macro-commande `convert`.

```
> y:=convert(f(x),piecewise);
```

(2.9)

$$y := \begin{cases} -3x+1 & x < \frac{1}{2} \\ x-1 & \frac{1}{2} \le x \end{cases} \quad (2.9)$$

Calculons donc y'. Nous allons utiliser la macro-commande `diff` car *y* ainsi formulée n'est pas (pour Maple) dans une relation fonctionnelle avec *x*.
Essayons voir.

```
> D(y);
```

$$\mathrm{D}\left(\begin{cases} -3x+1 & x < \frac{1}{2} \\ x-1 & \frac{1}{2} \le x \end{cases}\right) \quad (2.10)$$

```
> `y'`:=diff(y,x);
```

$$y' := \begin{cases} -3 & x < \frac{1}{2} \\ \mathit{undefined} & x = \frac{1}{2} \\ 1 & \frac{1}{2} < x \end{cases} \quad (2.11)$$

Puisque le domaine de la fonction f est $\mathbb{R}$, et que la fonction f n'est pas dérivable seulement en $x = \frac{1}{2}$, le domaine de $f'(x)$ est $\mathbb{R} \setminus \left\{\frac{1}{2}\right\}$.

Superposons, sur l'intervalle $x \in]-3, 3[$, le tracé de la fonction et celui de sa dérivée. Traçons en couleur `navy` la fonction f et en couleur "`Burgundy`" la fonction dérivée $f'(x)$. De plus, avec le tracé de deux petits cercle, illustrons que la fonction dérivée $f'(x)$ n'est pas définie en $x = \frac{1}{2}$.

```
> Courbe:=plot([x,y,x=-3..3],color=navy):
  Courbe_dérivée:=plot([x,`y'`,x=-3..4],color="Burgundy",discont=true):
  Trous:=plot({[1/2,-3],[1/2,1]},style=point,symbolsize=18,
                  symbol=circle,color="Burgundy"):
  display([Courbe,Courbe_dérivée,Trous],view=[-3..3,-4..3],background=
  Fond);
```



Exemple 1

Soit maintenant la fonction f définie par $f(x) = |x^2 - 5x + 6|$.

La fonction f est une fonction continue sur tous les réels $\mathbb{R}$. Traçons son graphique pour $x \in [0, 5]$. Contrôlons l'affichage des axes avec l'option `view=[0..5,-1..3]`.

```
> f:=x->abs(x^2-5*x+6);
  plot([x,f(x),x=0..5],scaling=constrained,view=[0..5,-1..3]);
```

$$f := x \rightarrow |x^2 - 5x + 6|$$



Calculons la fonction dérivée f'

```
> Diff(f(x),x)=D(f)(x);
```

$$\frac{d}{dx} |x^2 - 5x + 6| = (2x - 5)\,\mathrm{abs}(1, x^2 - 5x + 6) \qquad \textbf{(2.1.1)}$$

Le résultat affiché pour f' est formulé avec la syntaxe Maple pour la dérivée d'une valeur absolue: la dérivée de $|x^2 - 5 \cdot x + 6|$ est $(2x - 5)$ mais conditionnellement au signe de la simplification de $|x^2 - 5 \cdot x + 6|$.

Pour simplifier, trouvons pour quelle(s) valeur(s) de x le calcul de $x^2 - 5x + 6$ s'annule.

```
>  Racines:=solve(f(x)=0,x);
```

$$Racines := 2, 3 \qquad \textbf{(2.1.2)}$$

Analysez donc la nature du point $(2, f(2))$ et $(3, f(3))$. Évaluons donc $f'(2)$ et $f'(3)$.

```
> D(f)(2);
  D(f)(3);
Error, (in simpl/abs) abs is not differentiable at 0
Error, (in simpl/abs) abs is not differentiable at 0
```

Nous devons donc calculer les dérivées directionnelles en $x = 2$ et en $x = 3$.

```
> `f '`[`-`](2)=limit(D(f)(x),x=2,left);
  `f '`[`+`](2)=limit(D(f)(x),x=2,right);
```

$$f'_{-}(2) = -1$$

$$f'_{+}(2) = 1 \qquad \textbf{(2.1.3)}$$

Donc, puisque $f'_{-}(2) \neq f'_{+}(2)$, le point $(2, f(2))$ est un point anguleux.

Similairement, déterminons la nature du point $(3, f(3))$.

```
> `f'`[`-`](3)=limit(D(f)(x),x=3,left);
  `f'`[`+`](3)=limit(D(f)(x),x=3,right);
```

$$f'_{-}(3) = -1$$

$$f'_{+}(3) = 1 \qquad \textbf{(2.1.4)}$$

Cela qui montre également que le point $(3, f(3))$ est un point anguleux.

Nous allons réécrire la valeur absolue de la règle $f(x)$ en un règle définie pas morceaux et la dérivée avec la macro-commande `diff`. Cela aurait permis de déduire d'une autre façon la nature de ces deux points.

```
> y:=convert(f(x),piecewise);
  `y'`:=diff(y,x);
```

$$y := \begin{cases} x^2 - 5x + 6 & x \leq 2 \\ -x^2 + 5x - 6 & x < 3 \\ x^2 - 5x + 6 & 3 \leq x \end{cases}$$

$$y' := \begin{cases} 2x - 5 & x < 2 \\ \mathit{undefined} & x = 2 \\ -2x + 5 & x < 3 \\ \mathit{undefined} & x = 3 \\ 2x - 5 & 3 < x \end{cases} \qquad \textbf{(2.1.5)}$$

Superposons les tracés de la fonction f avec celui de sa dérivée. Nous allons aussi illustrer avec de petits cercles que la fonction dérivée n'est pas définie en $x = 2$ et en $x = 3$, $f'(2)$ et $f'(3)$ n'existe pas

```
> Courbes:=plot([[x,y,x=0..4],[x,`y'`,x=0..4]],discont=true,
          color=[navy,"Burgundy"]):
  Trous:=plot({[2,-1],[2,1],[3,-1],[3,1]},style=point,symbolsize=15,
          symbol=circle,color="Burgundy"):
  display(Courbes,Trous,background=Fond,view=[0..4,-3..3],size=[300,
```

```
300]);
```



Nous allons maintenant créer une petite animation illustrant les positions limites des droites sécantes passant par le point $(3, f(3))$.

Initialisons les éléments nécessaires à l'animation .

```
> h:='h':
  P:=[3,f(3)]:
  Q:=[P[1]+h,f(P[1]+h)]:
  TVM:=(P,Q)->(Q[2]-P[2])/(Q[1]-P[1]):
  Delta1:=[2,1.9,1.8,1.7,1.6,1.5,1,.9,.8,.7,.6,.5,.4,.3,.2,.1,
           .01,0.001,.0001,.00001]:
  Delta2:=map(x->-x,Delta1):
  Éq_Sécante:=y=TVM(P,Q)*(x-P[1])+P[2]:
```

Créons ensuite les structures graphiques qui seront superposées pour obtenir la petite animation

```
> Sécantes_D:=seq(plot([[x,rhs(Éq_Sécante),x=2.8..5],[x,f(x),x=0..5]
  ],
                 color=[khaki,navy]),h=Delta1):
  Sécantes_G:=seq(plot([[x,rhs(Éq_Sécante),x=0..3.2],[x,f(x),x=0..4]
  ],
                 color=[khaki,navy]),h=Delta2):
  Points_D:=seq(plot([P,Q],style=point,symbol=solidcircle,color=
  navy,symbolsize=15),
                 h=Delta1):
  Points_G:=seq(plot([P,Q],style=point,symbol=solidcircle,color=
  navy,symbolsize=15),
                 h=Delta2):
  A:=display(Sécantes_D,insequence=true):
  B:=display(Sécantes_G,insequence=true):
  C:=display(Points_D,insequence=true):
  E:=display(Points_G,insequence=true):
  display(A,B,C,E,view=[-1..5,-.5..6],background=Fond,size=[300,300]
  );
```



Après avoir exécutée cette animation, nous constaterons que les positions limites des sécantes à la droite et à la gauche du point $(3, f(3))$ plutôt que d'être confondues en une unique tangente forme alors un angle, d'où le nom point anguleux.

Exemple 2

Il n'y a pas seulement que les fonctions définies avec la valeur absolue qui peuvent présenter des points anguleux.

Considérons la fonction f définie par $f(x) = \sqrt{\frac{x^2(2-x)}{1+x}}$.

Il est certain que $x = -1$ ne fait pas parti du domaine de la fonction f. Mais en plus, on doit s'assurer que le calcul de $\frac{x^2(2-x)}{1+x} \geq 0$.

```
> solve(x^2*(2-x)/(1+x)>=0);
```

$$RealRange(Open(-1), 2) \quad \textbf{(2.2.1)}$$

Le domaine de la fonction f est donc intervalle $]-1, 2]$. Tracez donc la fonction f sur cette intervalle. Le dénominateur de la fonction f s'annulant près de -1, contrôlons l'affichage des axes en précisant l'option `view=[-1..3,-1..3]`.

```
> f:=x->sqrt(x^2*(2-x)/(1+x));
  plot([x,f(x),x=-1..2],color=navy,scaling=constrained,view=[-1..3,
  -1..3],size=[300,300]);
```

$$f := x \rightarrow \sqrt{\frac{x^2(2-x)}{x+1}}$$



Le domaine de la fonction f étant $]-1, 2]$, f n'est pas dérivable en $x = 2$. De plus, le graphique semble montrer qu'elle est dérivable partout sur $]-1, 2[$ sauf en $x = 0$. On dirait que le point $(0, f(0))$ est un point anguleux.

Calculons donc $f'(0)$.

```
> D(f)(0);
```

Error, (in anonymous procedure) numeric exception: division by zero

Donc, $f''(0)$ n'est pas définie. Le sens du message d'erreur est le suivant: l'évaluation de la dérivée amène une division par 0.

Calculons donc les dérivées à gauche et à droite en $x = 0$.

```
> `f '`[`-`](0)=limit(D(f)(x),x=0,left);
  `f '`[`+`](0)=limit(D(f)(x),x=0,right);
```

$$f'_{-}(0) = -\sqrt{2}$$

$$f'_{+}(0) = \sqrt{2} \quad \textbf{(2.2.2)}$$

Le point $(0, f(0))$ est bien une point anguleux.

Mais quelles sont les coordonnées de ce point anguleux ?

```
> Point_anguleux:=[0,f(0)];
```

$$Point\_anguleux := [0, 0] \quad \textbf{(2.2.3)}$$

Superposons les tracés des deux droites tangentes directionnelles avec celui de la fonction f au point $(0, 0)$. Obtenons les équations de ces deux tangentes directionnelles (gauche et droite).

```
> y:='y':
  P:=[0,0];
  m_g:=limit(D(f)(x),x=0,left);
  m_d:=limit(D(f)(x),x=0,right);
  Éq_tg_g:=y=m_g*(x-P[1])+P[2];
  Éq_tg_d:=y=m_d*(x-P[1])+P[2];
```

$$P := [0, 0]$$

$$m\_g := -\sqrt{2}$$

$$m\_d := \sqrt{2}$$

$$Éq\_tg\_g := y = -\sqrt{2}\, x$$

$$\acute{E}q\_tg\_d := y = \sqrt{2}\, x \tag{2.2.4}$$

Reste à créer la superposition demandée.

```
> Tangente_gauche:=plot([x,rhs(Éq_tg_g),x=-1..2],color="Burgundy"):
  Tangente_droite:=plot([x,rhs(Éq_tg_d),x=-1..1],color="Burgundy"):
  Courbe:=plot([x,f(x),x=-1..2],color=navy):
  display({Courbe,Tangente_gauche,Tangente_droite},
           scaling=constrained,
           size=[300,300],view=[-1..3,-1..3]);
```



Superposons les tracés de la fonction f et de sa dérivée f'. Illustrons, bien sûr, avec de petits cercles, que la fonction dérivée n'est pas définie en $x = 0$.

Donnez la couleur `navy` à la courbe f et la couleur `"Burgundy"` à la courbe f'. N'oublions pas l'option `discont=true` pour que le tracé de la fonction dérivée puisse être fait avec plus de discernement en $x = 0$.

```
> Courbes:=plot([[x,f(x),x=-1..2],[x,D(f)(x),x=-1..3]],discont=true,
  color=[navy,"Burgundy"]):
  Points:=plot([[0,-sqrt(2)],[0,sqrt(2)]],style=point,symbol=circle,
  color="Burgundy"):
  display({Courbes,Points},view=[-2..2,-4..4],background=Fond,size=
  [300,300]);
```



Les développements ci-dessous ont pour but d'introduire le lecteur à de nouvelles macro-commandes de "simplification". Comme vous le savez déjà, en mode manus scriptus, la simplification algébrique est exigeante. On y arrive étape par étape. Il en est de même avec Maple.

```
> `y'`:=D(f)(x);
```

$$y' := \frac{1}{2} \frac{\frac{2x(2-x)}{x+1} - \frac{x^2}{x+1} - \frac{x^2(2-x)}{(x+1)^2}}{\sqrt{\frac{x^2(2-x)}{x+1}}} \tag{2.2.5}$$

Normalisons cette fraction.

```
> Simp1:=normal((2.2.5));
```

$$Simp1 := -\frac{1}{2} \frac{x\left(2x^2+x-4\right)}{(x+1)^2 \sqrt{-\frac{x^2(x-2)}{x+1}}} \tag{2.2.6}$$

Simplifions le carré x^2 sous le radical.

```
> Simp2:=simplify(Simp1,sqrt,assume=real);
```

$$Simp2 := \frac{1}{2} \frac{-2x^2-x+4}{(x+1)^2 \operatorname{signum}(x) \sqrt{\frac{2-x}{x+1}}} \tag{2.2.7}$$

Convertissons signum(x) dans une notation valeur absolue.

```
> Simp3:=convert(Simp2,abs);
```

$$Simp3 := \frac{1}{2} \frac{\left(-2x^2-x+4\right)|x|}{(x+1)^2 x \sqrt{\frac{2-x}{x+1}}} \tag{2.2.8}$$

Comme dernière simplification, celle de la puissance de $1+x$, indiquons à l'évaluateur les restrictions suivantes sur x afin d'effectuer une dernière simplification sur le domaine de la fonction f.

```
> assume(x>-1);
  additionally(x<2);
```

Réduisons maintenant les puissance de $1+x$.

```
> `y'`:=simplify(Simp3,power);
```

$$y' := -\frac{1}{2} \frac{\left(2x^2+x-4\right)|x|}{(x+1)^{3/2} x \sqrt{2-x}}$$

With assumptions on x **(2.2.9)**

Interrogons l'évaluateur quant à la nature de ces restrictions.

```
> about(x);
Originally x, renamed x~:
  is assumed to be: RealRange(Open(-1),Open(2))
```

Ces restrictions concernent seulement celles qui ont été appliquées par l'évaluateur pour la dernière simplification. Rien de plus, rien de moins.

Avant de terminer cet exemple avec un dernier développement, rendez à nouveau la variable x "inconditionnellement mathématique" (rendre libre la variable x).

```
> x:='x':
```

Exemple 3

Voici un exemple d'une fonction continue sur $\mathbb{R}$ mais présentant un nombre infini de points anguleux.

Soit f la fonction définie par $f(x) = \left|\sin(x) + \frac{1}{2}\cos(x)\right|$. Le domaine de cette fonction est bien sûr $\mathbb{R}$.

Traçons la fonction f pour $[-2\pi..2\pi]$.

```
> f:=x->abs(sin(x)+1/2*cos(x));
  plot([x,f(x),x=-2*Pi..2*Pi],view=[-2*Pi..2*Pi,-1..1.2],color=navy)
  ;
```

$$f := x \rightarrow \left|\sin(x) + \frac{1}{2}\cos(x)\right|$$



Pour déterminer les coordonnées des "pics", où il nous semble que la fonction f ne soit pas dérivable, obtenons tous les zéros de cette fonction. Pour imposer à l'afficheur de retourner tout l'ensemble solution que l'évaluateur obtient à l'aide des réciproques des fonctions transcendantes, il faut initialiser la variable d'environnement `\_EnvAllSolutions` à la valeur de vérité `true`.

```
> _EnvAllSolutions:=true;
  solve(f(x));
```

$$\_EnvAllSolutions := true$$

$$-\arctan\left(\frac{1}{2}\right) + \pi\_Z3$$

With assumptions on _Z3 **(2.3.1)**

Maple informe de cette façon que les valeurs que doit prendre la variable système $\_Z_3$ sont des entiers.

L'ensemble-solution est donc $\left\{-\arctan\left(\frac{1}{2}\right) + k\pi \;\middle|\; k \in \mathbb{Z}\right\}$.

REMARQUE: Le choix de la notation $\_Z_3$ par l'évaluateur dépend de l'état du système au moment de la requête `solve`.

Choisissons une valeur particulière de la variable $\_Z_3$, pour un clacul de la dérivée de f.. Calculons la dérivée de la fonction f au point $\left(-\arctan\left(\frac{1}{2}\right), f\left(-\arctan\left(\frac{1}{2}\right)\right)\right)$.

Calculons $f'\left(-\arctan\left(\frac{1}{2}\right)\right)$.

```
> D(f)(-arctan(1/2));
Error, (in simpl/abs) abs is not differentiable at 0
```

Clairement. la fonction f n'est pas dérivable en $x = -\arctan\left(\frac{1}{2}\right)$.

Analysons une peu plus la situation en n $x = -\arctan\left(\frac{1}{2}\right)$. Calculons les dérivées à gauche et à droite $x = -\arctan\left(\frac{1}{2}\right)$.

```
> `f'`[`-`](-arctan(1/2))=limit(D(f)(x),x=-arctan(1/2),left);
  `f'`[`+`](-arctan(1/2))=limit(D(f)(x),x=-arctan(1/2),right);
```

$$f'_{-}\left(-\arctan\left(\frac{1}{2}\right)\right) = -\frac{1}{2}\sqrt{5}$$

$$f'_{+}\left(-\arctan\left(\frac{1}{2}\right)\right) = \frac{1}{2}\sqrt{5} \qquad \textbf{(2.3.2)}$$

Ce qui montre que le point $\left(-\arctan\left(\frac{1}{2}\right), f\left(-\arctan\left(\frac{1}{2}\right)\right)\right)$, c'est-à-dire le point $\left(-\arctan\left(\frac{1}{2}\right), 0\right)$ est un point anguleux.

Il est vraisemblable qu'il en est de même pour tout autre point d'abscisse $x \in \left\{-\arctan\left(\frac{1}{2}\right) + k\pi \mid k \in \mathbb{Z}\right\}$.

Superposons les tracés des deux tangentes directionnelles avec celui de la fonction f. Obtenons d'abord les équations de ces deux tangentes directionnelles.

```
> P:=[-arctan(1/2),f(-arctan(1/2))];
  m_g:=limit(D(f)(x),x=-arctan(1/2),left);
  m_d:=limit(D(f)(x),x=-arctan(1/2),right);
  Éq_tg_g:=expand(y=m_g*(x-P[1])+P[2]);
  Éq_tg_d:=expand(y=m_d*(x-P[1])+P[2]);
```

$$P := \left[-\arctan\left(\frac{1}{2}\right), 0\right]$$

$$m\_g := -\frac{1}{2}\sqrt{5}$$

$$m\_d := \frac{1}{2}\sqrt{5}$$

$$\acute{E}q\_tg\_g := y = -\frac{1}{2}\sqrt{5}\,x - \frac{1}{2}\sqrt{5}\arctan\left(\frac{1}{2}\right)$$

$$\acute{E}q\_tg\_d := y = \frac{1}{2}\sqrt{5}\,x + \frac{1}{2}\sqrt{5}\arctan\left(\frac{1}{2}\right) \qquad \textbf{(2.3.3)}$$

Créez maintenant la superposition demandée

```
> Tangente_gauche:=plot([x,rhs(Éq_tg_g),x=-2*Pi..2*
  Pi],color="Burgundy"):
  Tangente_droite:=plot([x,rhs(Éq_tg_d),x=-2*Pi..2*
  Pi],color="Burgundy"):
  Courbe:=plot([x,f(x),x=-2*Pi..2*Pi],color=navy):
  display({Courbe,Tangente_gauche,Tangente_droite},background=Fond,
  size=[300,300],view=[-2*Pi..2*Pi,-1..2]);
```




```
> display({Courbe,Tangente_gauche,Tangente_droite},background=Fond,
  size=[300,300],view=[-Pi/2..Pi/2,-1..2]);
```



Point de rebroussement

Nous allons analyser un autre type de "pic" que l'on peut rencontrer dans des graphiques.

Soit la fonction f définie par $x \mapsto \sqrt[3]{(3+x)\cdot(1-x)^2}$.

Le domaine de la fonction f est $\mathbb{R}$. Traçons le graphique de la fonction f pour $x \in [-5, 3]$.

Créons évidemment la fonction f avec la macro-commande `surd` puisqu'il s'agit de faire une analyse réelle de cette fonction.

```
> f:=x-> surd((3+x)*(1-x)^2,3);
  plot([x,f(x),x=-5..3],color=navy,scaling=constrained,view=[-5..3,-4.
```

```
.3]);
```

$$f := x \rightarrow \sqrt[3]{(3+x)\,(1-x)^2}$$



Obtenons la dérivée de la fonction f.

```
> Diff( 'f'(x),x)=D(f)(x);
```

$$\frac{\mathrm{d}}{\mathrm{d}x} f(x) = \frac{1}{3} \, \frac{\left((1-x)^2 - 2\,(3+x)\,(1-x)\right) \sqrt[3]{(3+x)\,(1-x)^2}}{(3+x)\,(1-x)^2} \qquad \textbf{(3.1)}$$

Simplifions.

Attention: Il revient à l'utilisateur d'établir le domaine de simplification. Il vous faudra évaluer si la perte d'information est significative ou non. Cette gestion passe évidemment par l'établissement, au départ, du domaine de définition de l'expression à simplifier.

```
> Diff( 'f'(x),x)=normal(rhs((3.1)));
```

$$\frac{\mathrm{d}}{\mathrm{d}x} f(x) = \frac{1}{3} \, \frac{\sqrt[3]{(3+x)\,(x-1)^2}\,(5+3x)}{(x-1)\,(3+x)} \qquad \textbf{(3.2)}$$

Calculons $f'(1)$.

```
> D(f)(1);
Error, (in anonymous procedure) numeric exception: division by zero
```

Il fallait s'y attendre. En effet, le domaine de la fonction dérivée est $\mathbb{R} \setminus \{-3, 1\}$. La fonction f n'est donc ni dérivable en $x = 1$ et ni en $x = 3$.

Il y a sur le graphique de la fonction f un "pic" au point $(1, f(1))$. Nous allons découvrir que la nature de ce pic n'est pas celle d'un point anguleux. Calculons les dérivées directionnelles $f'_{-}(1)$ et $f'_{+}(1)$.

```
> `f '`[`-`](1)=limit(D(f)(x),x=1,left);
  `f '`[`+`](1)=limit(D(f)(x),x=1,right);
```

$$f'_{-}(1) = -\infty$$

$$f'_{+}(1) = \infty \qquad \textbf{(3.3)}$$

Puisque la limite à gauche n'existe pas en $x = 1$, la fonction f n'est pas, effectivement dérivable en $x = 1$. Puisque la limite à droite n'existe pas non plus, la définition d'un point anguleux n'est pas ici satisfaite. Ce point de la fonction n'est donc pas un point anguleux.

Comment peut-on interpréter graphiquement ce point où les pentes des deux droites tangentes directionnelles sont de l'ordre de l'infini ?

Réalisons donc l'animation suivante qui illustrera la position limite des droites sécantes directionnelles passant par le point $(1, f(1)) = (1, 0)$.

Initialisons les éléments nécessaires à l'animation. Faisons un copier-coller des requêtes de l'animation de l'exemple 1 de la section « Point anguleux » et modifions ensuite les éléments pertinents.

```
> h:='h':
  P:=[1,f(1)]:
  Q:=[P[1]+h,f(P[1]+h)]:
  TVM:=(P,Q)->(Q[2]-P[2])/(Q[1]-P[1]):
  Delta1:=[3,2.5,2,1.9,1.8,1.7,1.6,1.5,1,.9,.8,.7,.6,.5,.4,.3,.2,.1,\
          .01,.001,.0001,.00001,0.000000001]:
  Delta2:=map(x->-x,Delta1):
  Éq_Sécante:=y=TVM(P,Q)*(x-P[1])+P[2]:
```

```
> Sécantes_D:=seq(plot([[x,rhs(Éq_Sécante),x=-1..4.5],[x,f(x),x=-6.
.4.5]],
                        color=[khaki,navy]),h=Delta1):
Sécantes_G:=seq(plot([[x,rhs(Éq_Sécante),x=-4..3],[x,f(x),x=-6..4.5]
],
                        color=[khaki,navy]),h=Delta2):
Points_D:=seq(plot([P,Q],style=point,symbol=solidcircle,color=navy,
symbolsize=15),
                h=Delta1):
Points_G:=seq(plot([P,Q],style=point,symbol=solidcircle,color=navy,
symbolsize=15),
                h=Delta2):
A:=display(Sécantes_D,insequence=true):
B:=display(Sécantes_G,insequence=true):
C:=display(Points_D,insequence=true):
E:=display(Points_G,insequence=true):
display(A,B,C,E,view=[-6..4,-4..4],scaling=constrained);
```



DÉFINITION

Un point P$(a, f(a))$ sur le graphique d'une fonction f est un **point de rebroussement** si f est continue en $x = a$ et si les deux conditions suivantes sont satisfaites:

– $f'(x) \to +\infty$ quand x s'approche de a d'un côté

– $f'(x) \to -\infty$ quand x s'approche de a de l'autre côté

Superposons les tracés de la fonction f et de la droite verticale d'équation $x = 1$.

```
> Courbe:=plot([x,f(x),x=-6..4],color=navy):
  Tg_Verticale:=plot([1,y,y=-5..4],color="Burgundy"):
  display({Courbe,Tg_Verticale},scaling=constrained,background=Fond,
  size=[300,300],view=[-6..4,-6..4]);
```



Vous avez peut-être remarqué que le tracé de la fonction f semble présenté un comportement asymptotique oblique.

Sans le document « Asymptotes_obliques_et_termes_dominants », Nousavons analyser le comportement asymptotique oblique dans le cas de fonctions rationnelles dont le degré du numérateur était supérieur d'un unité à celui du dénominateur. Dans de tels cas, l'équation de l'asymptote oblique est obtenu par le dividende de la division des deux polynômes (s'il y a un reste non nul tendant vers 0).

Dans le cas de la fonction f que l'on analyse, il faut procéder autrement pour montrer l'existence ou non d'un comportement asymptotique oblique.

Afin de peaufiner le tracé de la fonction f avec le tracé de l'asymptote oblique, il nous faut obtenir l'équation de cette asymptote *oblique*.

Pour obtenir les équations des asymptotes obliques, lorsqu'une courbe en possèdent, il faut:

1 remplacer y par $mx+b$ dans l'équation de la courbe et mettre le résultat sous la forme
$a_0 x^n + a_1 x^{n-1} + a_2 x^{n-2} + \ldots + a_{n-2} x^2 + a_{n-1} x + a_n$

2 résoudre simultanément les équations $a_0 = 0$ et $a_1 = 0$ pour m et b.

3 pour chaque couple de solutions m et b, écrire l'équation d'une asymptote $y = mx + b$

Si $a_1 = 0$, indépendamment de la valeur de b, les équations $a_0 = 0$ et $a_2 = 0$ doivent être utilisées dans (3).

Obtenons la reformulation de $f(x)$ en notation puissance fractionnaire.

```
> Équation:=y=convert(f(x),power);
```

$$Équation := y = \left((3+x)\,(1-x)^2\right)^{1/3} \tag{3.4}$$

Élevons chaque membre au cube. La macro-commande `map` permet d'appliquer une procédure à chaque opérande d'une expression.

```
> map(x->x^3,Équation);
```

$$y^3 = (3+x)\,(1-x)^2 \tag{3.5}$$

Remplaçons *y* par *mx*+*b* dans l'équation précédente. La macro-commande `subs` permet la substitution syntaxique d'une sous-expression correspondant à une opérande d'une autre expression.

```
> subs(y=m*x+b,(3.5));
```

$$(m\,x+b)^3 = (3+x)\,(1-x)^2 \tag{3.6}$$

Développons chaque membre. La macro-commande `expand` développe la distributivité de la multiplication sur l'addition.

```
> Équation_2:=expand((3.6));
```

$$Équation\_2 := m^3\,x^3 + 3\,b\,m^2\,x^2 + 3\,b^2\,m\,x + b^3 = x^3 + x^2 - 5\,x + 3 \tag{3.7}$$

Rendons nul le membre de droite en soustrayant membre à membre le membre de droite.

```
> Équation_3:=Équation_2-(rhs(Équation_2)=rhs(Équation_2));
```

$$Équation\_3 := m^3\,x^3 + 3\,b\,m^2\,x^2 + 3\,b^2\,m\,x + b^3 - x^3 - x^2 + 5\,x - 3 = 0 \tag{3.8}$$

Obtenons la forme $a_0 x^n + a_1 x^{n-1} + a_2 x^{n-2} + \ \ldots\ + a_{n-2} x^2 + a_{n-1} x + a_n$. La macro-commande `collect` permet de structurer une expression en pseudo-polynôme sur la base d'une de ses indéterminées.

```
> Équation_4:=collect(Équation_3,x);
```

$$Équation\_4 := \left(m^3-1\right) x^3 + \left(3\,b\,m^2-1\right) x^2 + \left(3\,b^2\,m+5\right) x + b^3 - 3 = 0 \tag{3.9}$$

Résolvons simultanément les équations $a_0 = 0$ et $a_1 = 0$ pour m et b.

```
> solve({coeff(lhs(Équation_4),x^3)=0,coeff(lhs(Équation_4),x^2)=0},{m,
  b});
```

$$\left\{b = \frac{1}{3}, m = 1\right\}, \left\{b = \frac{1}{3}\,RootOf\left(\_Z^2 + \_Z + 1\right), m = RootOf\left(\_Z^2 + \_Z + 1\right)\right\} \tag{3.10}$$

Nous allons, bien sûr, ne retenir que la solution réelle. L'équation cherchée de l'asymptote oblique est donc $y = x + \frac{1}{3}$.

```
> Asymptote_Oblique:=plot([x,x+1/3,x=-5..3],linestyle=3,color=khaki):
  display({Courbe,Asymptote_Oblique},scaling=constrained,view=[-5..3,
  -4..3]);
```



Montrons que $\lim_{x \to \pm\infty} \left(\sqrt[3]{(3+x)(1-x)^2} - x - \frac{1}{3} \right) = 0$. Cela attestera le comportement asymptotique de la fonction f et et de la droite $y = x + \frac{1}{3}$.

```
> Limit(f(x)-(x+1/3),x=infinity)=limit(f(x)-(x+1/3),x=infinity);
  Limit(f(x)-(x+1/3),x=-infinity)=limit(f(x)-(x+1/3),x=-infinity);
```

$$\lim_{x \to \infty} \left(\sqrt[3]{(3+x)(1-x)^2} - x - \frac{1}{3} \right) = 0$$

$$\lim_{x \to -\infty} \left(\sqrt[3]{(3+x)(1-x)^2} - x - \frac{1}{3} \right) = 0 \quad \textbf{(3.11)}$$

Point de tangence verticale

Considérons, à nouveau, la fonction f définie par $f(x) = \sqrt[3]{(3+x)(1-x)^2}$ et sa dérivée obtenu précédemment.

```
> `y'`:= rhs((3.2));
```

$$y' := \frac{1}{3} \frac{\sqrt[3]{(3+x)(x-1)^2}\,(5+3x)}{(x-1)(3+x)} \quad \textbf{(4.1)}$$

Affichons directement la simplification de la dérivée $f'(x)$ obtenue précédemment.

```
> `` =convert((4.1),power);
```

$$= \frac{1}{3} \frac{\left((3+x)(x-1)^2\right)^{1/3}(5+3x)}{(x-1)(3+x)} \quad \textbf{(4.2)}$$

```
> ``=simplify(rhs((4.2)),symbolic);
```

$$= \frac{1}{3} \frac{5+3x}{(3+x)^{2/3}(x-1)^{1/3}} \quad \textbf{(4.3)}$$

Cela montre également que la fonction f n'est pas dérivlable $x = -3$.

Cette situation de non dérivabilité est différente des deux cas précédemment étudiés. Graphiquement, en $x = -3$, il ne se présente ni un point anguleux ni un point de rebroussement.

Nous allons aussi étudier la nature de cette situation de non dérivabilité en calculant les dérivées directionnelles à gauche et à droite en $x = -3$.

```
> `f '`[`-`](-3)=limit(D(f)(x),x=-3,left);
  `f '`[`+`](-3)=limit(D(f)(x),x=-3,right);
```

$$f'_{-}(-3) = \infty$$

$$f'_{+}(-3) = \infty \tag{4.4}$$

Cette fois-ci, les deux dérivées directionnelles en $x = -3$ se comportent d'une manière semblable. Nous allons maintenant créer une animation illustrant les positions limites des droites sécantes à la courbe passant par le point $(1, f(1))$. Non seulement nous allons réaliser que c'est la même position verticale limite mais que, contrairement à la situation d'un point de rebroussement, cette position limite est atteinte, dans les deux cas, avec une même direction (dans ce cas-ci positive) de toutes les droites sécantes de l'animation.

Initialisons les éléments nécessaires à l'animation.

```
> h:='h':
  P:=[-3,f(-3)]:
  Q:=[P[1]+h,f(P[1]+h)]:
  TVM:=(P,Q)->(Q[2]-P[2])/(Q[1]-P[1]):
  Delta1:=[3,2.5,2,1.9,1.8,1.7,1.6,1.5,1,.9,.8,.7,.6,.5,.4,.3,.2,.1,\
           .01,.001,.0001,.00001,0.000000001]:
  Delta2:=map(x->-x,Delta1):
  Éq_Sécante:=y=TVM(P,Q)*(x-P[1])+P[2]:
```

```
> Sécantes_D:=seq(plot([[x,rhs(Éq_Sécante),x=-5..0],[x,f(x),x=-6..4]],
                  color=[khaki,navy]),h=Delta1):
  Sécantes_G:=seq(plot([[x,rhs(Éq_Sécante),x=-6..-1],[x,f(x),x=-6..4]],
                  color=[khaki,navy]),h=Delta2):
  Points_D:=seq(plot([P,Q],s,style=point,symbol=solidcircle,color=navy,
  symbolsize=15),
                h=Delta1):
  Points_G:=seq(plot([P,Q],style=point,symbol=solidcircle,color=navy,
  symbolsize=15),
                h=Delta2):
  A:=display(Sécantes_D,insequence=true):
  B:=display(Sécantes_G,insequence=true):
  C:=display(Points_D,insequence=true):
  E:=display(Points_G,insequence=true):
  display({A,B,C,E},view=[-6..4,-6..6],scaling=constrained);
```



Après avoir exécuté cette animation, on constate que les positions limites des droites sécantes à la droite et à la gauche du point $(3, f(3))$ sont confondues en une unique droite tangente verticale obtenue avec une même direction (positive ici) des différentes droites sécantes à gauche et à droite de $x = -3$.

DÉFINITION

Le graphique d'une fonction f a une tangente verticale d'équation $x = a$ au point $P(a, f(a))$ si f est continue en a et si

$$\lim_{x \to a} |f'(x)| = \infty$$

Dans ce cas-ci (en $x = -3$) comme dans le cas d'un point de rebroussement (en $x = 1$), il y a une tangente verticale et donc que la définition précédente est satisfaite. Mais, au point $(-3, 0)$, les deux conditions de la définition d'un point de rebroussement ne sont pas satisfaites.

Pour terminer cet exemple, superposons les tracés de la fonction f et de la fonction dérivée. Notons la différence de comportement du tracé de la fonction dérivée au voisinage des valeurs de x où les tangentes à la courbe f sont verticales.

```
> Courbe:=plot([x,f(x),x=-5..3],color=navy):
  Courbe_d:=plot([x,D(f)(x),x=-5..3],color="Burgundy",discont=true):
  display({Courbe,Courbe_d},scaling=constrained,background=Fond,size=
[300,300],view=[-5..3,-4..4]);
```



Lorsque le point P est une extrémité du graphique, la définition précédente reste valable si l'on ne considère, selon l'extrémité, qu'une seule dérivée directionnelle: la dérivée à gauche ou la dérivée à droite selon le cas.

À l'exemple 2 de la section « Point anguleux », nous avons établi que la fonction f définie par

$f(x) = \sqrt{\frac{x^2 (2-x)}{1+x}}$ possédait un point anguleux $(0, 0)$ et que la fonction, bien que continue à gauche en $x = 2$, n'était pas dérivable à gauche en $x = 2$.

Montrons qu'en $x = 2$ il y a une tangente verticale en alculant la dérivée à gauche.

```
> f:=x->sqrt(x^2*(2-x)/(1+x));
  `f'`[`-`](2)=limit(D(f)(x),x=2,left);
```

$$f := x \rightarrow \sqrt{\frac{x^2 (2-x)}{x+1}}$$

$$f'_{-}(2) = -\infty \qquad \textbf{(4.5)}$$

```
> Courbe:=plot([x,f(x),x=-1..2],color=navy):
Tg_Verticale:=plot([2,y,y=-3..3],color="Burgundy"):
display({Courbe,Tg_Verticale},scaling=constrained,background=Fond,
size=[300,300],view=[-1..3,-1..3]);
```